Algunos aspectos de polinomios de Bernstein, Bezier y trazadores

Vernor Arguedas Troyo
Roberto Mata Montero
Escuela de Matemática
Universidad de Costa Rica

Resumen

Este trabajo consta de dos partes: la primera presenta, de manera elemental, la teoría de los polinomios de Bernstein en una variable; la segunda esta dedicada a curvas de Bezier y q-trazadores ("q-splines"). Nos parece importante el uso que se puede dar del software *Mathematica*.

Palabras claves: aproximación, interpolación, splines, curvas, *Mathematica*.

Introducción

Como introducción del trabajo que desarrollaremos posteriormente, recordemos la definición de polinomios de Bernstein en su forma usual.

Definición [Polinomios de Bernstein]

Sea $f:[0,1]\longrightarrow\mathbb{R}$ una función continua. El polinomio n-ésimo de Bernstein $B(n,f,x)$ de la función f se define como:

$$B(n,f,x)=B_n(f,x)=\sum_{k=0}^{n} f\left(\frac{k}{n}\right)\binom{n}{k}x^k(1-x)^{n-k},$$

donde

$$\binom{n}{k}=\frac{n!}{k!\,(n-k)!}.$$

Una construción de los polinomios de Bernstein para la función constante $f(x)=1$ sobre el intervalo $[0,1]$, la hacemos de la siguiente forma

Observemos que $1^n=1$, $\forall\ n\in\mathbb{N}$. Por lo tanto, $\forall\ x\in\mathbb{R}$ se cumple que:

$$1=(x+1-x)^n=\sum_{k=0}^{n}\binom{n}{k}x^k(1-x)^{n-k}$$

lo cual nos dice que el n-ésimo polinomio de Berntein de la función constante $f(x)=1$ coincide con f, o sea es 1.

Con esto como preámbulo y asumiendo que $f:[0,1]\longrightarrow\mathbb{R}$ es continua, definimos en *Mathematica* el n-ésimo polinomio de Bernstein evaluado en x de la siguiente forma:

In[1]:=

```
B[n_, f_, x_] := Sum[Binomial[n, k] x^k (1 - x) ^ (n - k) f[k / n], {k, 0, n}];
```

$$B[n\_,\ f\_,\ x\_] := \sum_{k=0}^{n} \text{Binomial}[n,\ k]\ x\text{^}k\ (1-x)\ \text{^}\ (n-k)\ f[k/n];$$

No deja de ser interesante que *Mathematica* acepta ese procedimento sin dificultad, logrando resultados como los siguiente:

In[2]:=

```
B[n, f, x] /. f -> (# &)
```

Out[2]=

$$(1-x)^n \left(-\frac{1}{-1+x}\right)^n x$$

Observemos que *Mathematica* no simplifica el resultado. Para lograr una expresión más simple debemos definir la siguiente regla lógica:

In[3]:=

$$\text{regla} = \left\{(1-x)^n \left(-\frac{1}{-1+x}\right)^n \to 1,\ (1-x)^n \left(\frac{1}{1-x}\right)^n \to 1\right\};$$

```
B[n, f, x] /. f -> (# &) /. regla
```

Out[4]=

$$x$$

Es decir, el polinomio n-ésimo de Bernstein de la función $f(x) = x$ coincide con f, o sea es x.

Veamos otros ejemplos en los cuales la función $f(x)$ esta definida por x^2, x^3, x^4, $3x^4 + x^2$, Sen(x) + $3x^4 + x^2$, respectivamente.

Para la función $f(x) = x^2$

In[5]:=

```
Apart[Simplify[Expand[B[n, f, x] /. f → (#^2 &)]] /. regla, x]
```

Out[5]=

$$\frac{x}{n} + \frac{(-1+n)\, x^2}{n}$$

obtenemos que el n-ésimo polinomio de Bernstein es:

$$B_n\ (x^2,\ x) = \frac{x}{n} + \frac{(-1+n)\ x^2}{n}$$

Para la función $f(x) = x^3$

In[6]:=

```
Apart[
 Simplify[Expand[B[n, f, x] /. f → (#^3 &)]] /. regla]
```

Out[6]=

$$\frac{x}{n^2} + \frac{3\ (-1+n)\ x^2}{n^2} + \frac{(2-3\,n+n^2)\ x^3}{n^2}$$

obtenemos que el n-ésimo polinomio de Bernstein es:

$$B_n(x^3, x) = \frac{x}{n^2} + \frac{3(-1+n)x^2}{n^2} + \frac{(2-3n+n^2)x^3}{n^2}$$

Para la función $f(x) = x^4$

In[7]:=

```
Apart[Simplify[Expand[B[n, f, x] /. f → (#^4 &)]] /. regla, x]
```

Out[7]=

$$\frac{x}{n^3} + \frac{7(-1+n)x^2}{n^3} + \frac{6(2-3n+n^2)x^3}{n^3} + \frac{(-6+11n-6n^2+n^3)x^4}{n^3}$$

obtenemos que el n-ésimo polinomio de Bernstein es:

$$B_n(x^4, x) = \frac{x}{n^3} + \frac{7(-1+n)x^2}{n^3} + \frac{6(2-3n+n^2)x^3}{n^3} + \frac{(-6+11n-6n^2+n^3)x^4}{n^3}$$

Para la función $f(x) = 3x^4 + x^2$ obtenemos que

In[8]:=

```
Apart[Simplify[Expand[B[n, f, x] /. f → (3*#^4 + #^2 &)]] /. regla,
 x]
```

Out[8]=

$$\frac{(3+n^2)x}{n^3} + \frac{(-21+21n-n^2+n^3)x^2}{n^3} + \frac{18(2-3n+n^2)x^3}{n^3} + \frac{3(-6+11n-6n^2+n^3)x^4}{n^3}$$

y para la función $f(x) = \mathrm{Sen}(x) + 3x^4 + x^2$ obtenemos que

In[9]:=

```
Apart[Simplify[Expand[B[n, f, x] /. f → (Sin[#] +3*#^4 + #^2 &)]] /.
  regla, x]
```

Out[9]=

$$\frac{(3+n^2)\left(\frac{1}{1-x}\right)^n (1-x)^n x}{n^3} + \frac{(-21+21n-n^2+n^3)\left(\frac{1}{1-x}\right)^n (1-x)^n x^2}{n^3} + \frac{18(2-3n+n^2)\left(\frac{1}{1-x}\right)^n (1-x)^n x^3}{n^3} + \frac{3(-6+11n-6n^2+n^3)\left(\frac{1}{1-x}\right)^n (1-x)^n x^4}{n^3} - \frac{1}{2} i (1-x)^n \left(\left(\frac{-1+x-e^{\frac{i}{n}}x}{-1+x}\right)^n - \left(\frac{1+\left(-1+e^{-\frac{i}{n}}\right)x}{1-x}\right)^n\right)$$

Observe el uso que se dió a la función pura (#&), si no se usan los paréntesis el procedimiento no funciona.

Los polinomios de Bernstein tiene las siguientes propiedades:

- Si $f : [0, 1] \longrightarrow \mathbb{R}$ es continua entonces $B[n, f, x] = B_n[f, x]$ es una función continua.
- Si $f \geqslant 0$ entonces $B[n, f, x] = B_n[f, x] \geqslant 0$, para toda $x \in \mathbb{R}$.

Además, en $C([0, 1], \| \|_\infty)$ se cumple que para toda $f \in C([0, 1])$ se tiene que $B[n, f, x] \longrightarrow f$ con la norma $\| \|_\infty$. Es decir, la convergencia es uniforme. La demostración se basa en los siguientes hechos:

Primero, como f es continua en $[0, 1]$ entonces f es acotada y podemos elegir $M = \operatorname{Sup}_{0 \leqslant x \leqslant 1} \{f(x)\}$, con lo cual

$-M \leqslant f(x) \leqslant f(x), \quad \forall\ x \in [0, 1]$

Por otro lado, al ser f continua en $[0, 1]$, para cualquier $\zeta > 0$ existe un $\delta > 0$ tal que para todo $x, y \in [0, 1]$, que cumplen $|x - y| < \delta$ se tine que $|f(x) - f(y)| < \zeta$

Con las definiciones anteriores de M, ζ, δ, podemos afirmar que:

$$-\zeta - 2 \frac{M}{\delta^2} (x - y)^2 \leqslant f(x) - f(y) \leqslant \zeta + 2 \frac{M}{\delta^2} (x - y)^2, \quad \forall\ x, y \in [0, 1]$$

Con esto, el resto de la demostración es sencilla. Esencialmente ésta es la demostración del teorema de Korovkin. (El Lic. Gerardo Araya escribió una excelente tesis de graduación sobre este tema, dirigida por el profesor Vernor Arguedas [1])

Es importante recalcar que los polinomios de Bernstein aproximan a la función f, no la interpolan necesariamente.

Los términos intermedios en la construcción del n-ésimo polinomio de Bernstein de la función $f(x)$:

$$B_2(i, n) = f\left(\frac{i}{n}\right) \binom{n}{i} x^i (1 - x)^{i-k}, \quad \text{para } i = 0, 1, \ldots, n$$

$$B_2(n) = \{B_2(i, n) \mid i = 0, 1, 2, \ldots, n\}$$

son muy útiles.

Usando *Mathematica* podemos graficar los términos intermedios del polinomio de Bernstein $B(4, 1, x)$ (observe que $f(x) = 1$):

In[60]:=

```
Bernstein2[i_, n_] := Binomial[n, i] x^i (1 - x)^(n-i);
Bernstein2[n_] := Table[Bernstein2[i, n], {i, 0, n}];
Print["Términos intermedios para n=4"];
Bernstein2[4]
n = 4;
Print["Gráfica de los términos intermedios"];
Plot[Evaluate[Bernstein2 [n], {x, 0, 1}, PlotStyle→ {
    {RGBColor[0, 0, 1], Thickness[0.007]},
    {RGBColor[0, 1, 0], Thickness[0.007] },
    {RGBColor[0, 1, 1], Thickness[0.007]},
    {RGBColor[1, 0, 0], Thickness[0.007]} ,
    {RGBColor[1, 1, 0], Thickness[0.007]}
   } ,
  Epilog→ {AbsolutePointSize[5],
    Table[Point[{i / n, Bernstein2[i, n] /. x → i / n}], {i, 0, n} ] }]];
```

Términos intermedios para n=4

Out[63]=

$$\left\{(1-x)^4,\ 4\,(1-x)^3\,x,\ 6\,(1-x)^2\,x^2,\ 4\,(1-x)\,x^3,\ x^4\right\}$$

Gráfica de los términos intermedios



Figura 1: Gráfica de los términos intermedios del 4-ésimo polinomio de Bernstein.

In[67]:=

```
Plot[Evaluate[Bernstein2 [n], {x, 0, 1}, PlotStyle→ {
    {RGBColor[0, 0, 1], Thickness[0.007]},
    {RGBColor[0, 1, 0], Thickness[0.007] },
    {RGBColor[0, 1, 1], Thickness[0.007]},
    {RGBColor[1, 0, 0], Thickness[0.007]} ,
    {RGBColor[1, 1, 0], Thickness[0.007]}
   } ,
  Epilog→ {AbsolutePointSize[7],
    Table[Point[{i / n, Bernstein2[i, n] /. x→i / n}], {i, 0, n} ] ,
    {RGBColor[1, 0, 1], Thickness[0.007],
     Line[Table[{i / n, Bernstein2[i, n] /. x→i / n}, {i, 0, n} ] ]}}]];
```



Figura 2: Términos intermedios de 4-ésimo polinomio de Bernstein.

Curvas de Bezier

Usando las definiciones anteriores podemos definir las curvas de Bezier directamente en *Mathematica*:

In[68]:=

```
Bezier[pts_] := Bernstein2[Length[pts] - 1].pts;
```

en donde **pts** es una familia finita de pares ordenados de $\mathbb{R}^2$. Además, en Bernstein2[Length[pts] − 1].pts, el punto (.) denota el producto matricial, aunque puede significar muchas cosas dependiendo de los operandos que usemos (multiplicación de matrices, producto escalar, etc.). En este caso es una multiplicación de matrices cuyo resultado es una matriz 1×2. Visualicemos esto mediante el siguiente ejemplo en el que se construye un as de espadas, aunque no del todo terminado.

In[213]:=

```
p1 = {{-0.144, -0.830}, {-0.069, -0.588}, {-0.037, -0.253}};
AsespadasPlot1 = Bezier[p1];
p2 = {{-0.060, -0.240}, {-0.233, -0.444}, {-0.498, -0.430},
  {-0.712, -0.156}, {-0.637, 0.160}, {-0.423, 0.462},
  {-0.149, 0.630}, {-0.056, 0.756}, {0, 0.853}};

AsespadasPlot2 = Bezier[p2];
Show[Graphics[{AbsolutePointSize[7], RGBColor[1, 0, 0],
   Point /@ Join[p1, p2]}], Axes → True, AspectRatio → Automatic] ;
```



Figura 3: As de espadas.

```
AsespadasPoints = Module[{LeftSide},
  LeftSide = Join[AsespadasPlot1[[1, -1, 1, 1]],
    AsespadasPlot2[[1, -1, 1, 1]]];
  Join[LeftSide, Reverse[Map[{-1, 1} * # &, LeftSide]]]];
Sow[Graphics[Polygon[AsespadasPoints]], AspectRatio → Automatic];
```

Si tenemos una función $f:[0,1] \longrightarrow \mathbb{R}$ y una familia finita de puntos I dada por

$I = \{x_1, x_2, \ldots, x_n\}$ donde $a = x_1 < x_2 < \ldots < x_n = b$

obtenemos una familia de puntos de la forma:

$\{(a, f(a)), (x_2, f(x_2)), \ldots, (b, f(b))\}$

Observemos con cuidado que hace *Bezier*, ejemplificado para el caso de una partición uniforme, por comodidad.

```
n = 5;
puntos = Table[{a + k * (b - a) / n, f[a + k * (b - a) / n]}, {k, 0, n}];
Bezier[puntos] // Expand;
%[[1]]
%%[[2]] // Simplify
```

```
a - a x + b x
```

$$-(-1+x)^5 f[a] + x\left(-10(-1+x)^3 x f\left[\frac{3a}{5}+\frac{2b}{5}\right] + 10(-1+x)^2 x^2 f\left[\frac{2a}{5}+\frac{3b}{5}\right] + x^4 f[b] + 5 f\left[\frac{1}{5}(4a+b)\right] - 20 x f\left[\frac{1}{5}(4a+b)\right] + 30x^2 f\left[\frac{1}{5}(4a+b)\right] - 20x^3 f\left[\frac{1}{5}(4a+b)\right] + 5x^4 f\left[\frac{1}{5}(4a+b)\right] + 5x^3 f\left[\frac{1}{5}(a+4b)\right] - 5x^4 f\left[\frac{1}{5}(a+4b)\right]\right)$$

La primera componente nos define una parametrización lineal de $[0,1]\longrightarrow[a,b]$, mientras que la segunda componente nos da $\text{Bernstein}[n, f[a + b(x-a)], x]$, para $x\in[0,1]$.

Definición [q-trazadores] Sea $p:[a,b]\longrightarrow\mathbb{R}$ una función continua. f se llama q-trazador ("q-spline"), si existe una partición ÷ de $[a,b]$, dada por

$$\div = \{a = x_1, x_2, ..., x_{n-1}, x_n = b\}$$

tal que $p \mid [x_i, x_{i+1}]$ (p restringida al subintervalo $[x_i, x_{i+1}]$) es un polinomio de grado q. Los puntos de ÷ se llaman los nodos del trazador. El trazador se llama trazador (q,r), ("(q,r)-spline"), si exigimos que $p\in C^r([a,b],\mathbb{R})$.

El teorema fundamental nos dice que si $f:[a,b]\longrightarrow\mathbb{R}$ es una función de clase C^r y $\mathcal{P}=\{a=x_1, x_2, \ldots, x_n=b\}$ es una partición de $[a,b]$, entonces existe un trazador (q,r) que interpola a f hasta el orden $r-1$, en los puntos $x_1, x_2, \ldots, x_n$ (en los puntos x_1 y x_n a veces se dan otras condiciones).

Mathematica tiene varios paquetes útiles como *Numerical Math 'SplineFit'* el cual es un programa no tan general como *Interpolating polynomial; es* una implementación en *Mathematica* del algoritmo presentado por Burden ([3]), para el caso de trazadores cúbicos. No esta diseñado para el caso n-dimensional, $n>2$.

Construyamos un ejemplo sencillo. Pero observemos los tipos de trazadores: Type-Cubic, CompositeBezier y Bezier.

In[186]:=

```
Clear[n, k];
puntos1 = Table[{-2 + k / n, 1 / (1 + 12 * (-2 + k / n))}, {k, 0, 6}] /. n → 6
```

Out[187]=

$$\left\{\left\{-2, -\frac{1}{23}\right\}, \left\{-\frac{11}{6}, -\frac{1}{21}\right\}, \left\{-\frac{5}{3}, -\frac{1}{19}\right\}, \left\{-\frac{3}{2}, -\frac{1}{17}\right\}, \left\{-\frac{4}{3}, -\frac{1}{15}\right\}, \left\{-\frac{7}{6}, -\frac{1}{13}\right\}, \left\{-1, -\frac{1}{11}\right\}\right\}$$

En dichos puntos estamos haciendo un muestreo de la función $f(x) = \frac{1}{1+12\,x}$ en $[-2, -1]$.

In[196]:=

```
<< NumericalMath`SplineFit`
gg = SplineFit[puntos1, Cubic]
gg[x]
Print["Primer valor de la tabla de puntos"];
gg[0]
Print["Sexto valor de la tabla de puntos"]
gg[6]
gg[5.25]
```

Out[197]=

```
SplineFunction[Cubic, {0., 6.}, <>]
```

Out[198]=

```
(SplineFunction[Cubic, {0., 6.}, <>])[x]
```

```
Primer valor de la tabla de puntos
```

Out[200]=

$$\left\{-2, -\frac{1}{23}\right\}$$

```
Sexto valor de la tabla de puntos
```

Out[202]=

$$\left\{-1, -\frac{1}{11}\right\}$$

Out[203]=

```
{-1.125, -0.0801397}
```

$gg[x]$ define una función cuyo dominio es un intervalo cerrado que inicia en el primer k de la tabla, en este caso 0, y finaliza en el último valor de k, en este caso 6.

$gg : [0, 6] \longrightarrow \mathbb{R}^2$ cuya segunda componente es la aproximación por trazadores cúbicos de f. Para que nos sea útil tenemos que convertirla en una función definida sobre el intervalo $[-2, -1]$ lo cual hacemos parametrizando linealmente; en este caso con $x \to 6\,(2+x)$.

Usando *Mathematica*

In[204]:=

```
ggg[x_] := gg[6 * (2 + x)][[2]];
Plot[{ggg[x], 1 / (1 + 12 x)}, {x, -2, -1},
  PlotStyle → {RGBColor[0, 0, 1]},
  Epilog → {PointSize[0.03],
    Table[Point[{-2 + k / n, 1 / (1 + 12 * (-2 + k / n))}], {k, 0, 6}] /. n → 6}];
```



Figura 4: Gráficas de las funciones ggg[x] y f[x].

Las gráficas de ambas funciones ggg$[x]$ y $f(x) = \dfrac{1}{1 + 12\,x}$ coinciden.

In[206]:=

```
hh[t_] := Bezier[puntos1] /. x → t;
hh[-2]
hh[0]
hh[1]
Clear[hhh];
hhh[x_] := hh[2 + x][[2]]
Plot[{ggg[x], 1 / (1 + 12 x), hhh[x]}, {x, -2, -1},
 PlotStyle → {RGBColor[0, 0, 1], RGBColor[0, 0, 1],
   RGBColor[1, 0, 0]}];
```

Out[207]=

$$\left\{-4,\ -\frac{154867}{7436429}\right\}$$

Out[208]=

$$\left\{-2,\ -\frac{1}{23}\right\}$$

Out[209]=

$$\left\{-1,\ -\frac{1}{11}\right\}$$



Figura 5: Gráficas de las funciones ggg[x], f[x] y hhh[x].

Las gráficas de color azul corresponde a las funciones $\mathbf{ggg}[x]$ y $f(x) = \dfrac{1}{1+12\,x}$, las cuales coinciden. La gráfica de color rojo corresponde a la función $\mathbf{hhh}[x]$, generada con *Bezier*.

Bibliografía

[1] Araya Aguilar, Gerardo. *Solución a algunos problemas de Korovkin*. (1982), UCR, Costa Rica.
[2] Borwein, P.; T. Erdelyi. *Polinomials and polinomial inequalities*. (1996), SIAM Review, vol 38, # 4. pp. 705-706.
[3] Burden, Richard; Douglas Faires. *Análisis numérico*. Editorial Iberoamérica, México, 1985.
[4] Chui, C. K.; L. Schumaker; J. D. Wardor. *Approximation theory V*. (1988). Mathematics of computatino, vol 50, # 181. pp. 356.
[5] Gamelin, T. W.; R. E. Greene. *Introduction to topology*. (1986), American Mathematical Monthly, vol. 93, # 7. pp. 580-582.
[6] Sendov, Blagovet; Vasil A. Popov; G. M. Phillips. The averaged moduli of smoothness:with applications in numerical methods and approximation. (1991) SIAM Review, vol. 33, #3. pp. 499-501.